# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=libretro-mesen-s
pkgver=916
pkgrel=3
pkgdesc='Super Nintendo Entertainment System core'
arch=(loong64 x86_64)
url=https://github.com/SourMesen/Mesen-S
license=(GPL3)
groups=(libretro)
depends=(
  gcc-libs
  libretro-core-info
)
makedepends=(git)
_commit=89199f298a95345a893d814525b6668567a6d490
source=(git+https://github.com/SourMesen/Mesen-S.git#commit=${_commit})
sha256sums=(SKIP)

pkgver() {
  cd Mesen-S

  git rev-list --count HEAD
}

build() {
  make -C Mesen-S/Libretro
}

package() {
  install -Dm 644 Mesen-S/Libretro/mesen-s_libretro.so -t "${pkgdir}"/usr/lib/libretro/
}

# vim: ts=2 sw=2 et:
